# $Id$
# Maintainer: Tobias Powalowski <tpowa@archlinux.org>

pkgname=openexr
pkgver=2.3.0
pkgrel=1
pkgdesc="An high dynamic-range image file format library"
url="http://www.openexr.com/"
arch=('x86_64')
license=('BSD')
depends=('zlib' 'ilmbase')
source=("https://github.com/openexr/openexr/releases/download/v$pkgver/$pkgname-$pkgver.tar.gz")
md5sums=('a157e8a46596bc185f2472a5a4682174')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -D -m644 LICENSE -t "${pkgdir}/usr/share/licenses/${pkgname}"
}
